



599 Menlo Drive, Suite 100
Rocklin, California 95765, USA
Office: (916) 624-8333
Fax: (916) 624-8003

General: info@parallaxinc.com
Technical: stamptech@parallaxinc.com
Web Site: www.parallaxinc.com
Educational: www.stampsinclass.com

Devantech SRF04 Ultrasonic Range Finder (#28015)

The Devantech SRF04 ultrasonic range finder provides precise, non-contact distance measurements from about 3 cm (1.2 inches) to 3 meters (3.3 yards). It is very easy to connect to BASIC Stamps or the Javelin, requiring only two I/O pins.¹ The SRF04 library makes this device very simple to use and is an ideal component for robotics applications.

The SRF04 works by transmitting an ultrasonic (well above human hearing range) pulse and measuring the time it takes to "hear" the pulse echo. Output from the SRF04 is in the form of a variable-width pulse that corresponds to the distance to the target.

The SRF04 is designed and manufactured by Devantech, who provides additional technical resources for the device. Their web site is <http://www.robot-electronics.co.uk>.


Features

- Voltage - 5v
- Current - 30mA Typ. 50mA Max.
- Frequency - 40KHz
- Max Range - 3 m
- Min Range - 3 cm
- Sensitivity - Detect 3cm diameter broom handle at > 2 m
- Input Trigger - 10uS Min. TTL level pulse
- Echo Pulse - Positive TTL level signal, width proportional to range.
- Small Size – (1.7 in x .8 in x .7 in height) 43mm x 20mm x 17mm height

¹ For a Javelin Stamp application note see www.javelinstamp.com.


Connection to the BASIC Stamp 2

The SRF04 has four through-hole locations where you will need to solder wires to the hardware. These wires are not included with the kit.



Source Code Example

The SRF04 detects objects by emitting a short burst of sound and "listening" for the echo. Under control of the BASIC Stamp, the SRF04 emits an ultrasonic (40 kHz) sound pulse. This pulse travels through the air at about 1.1 feet per millisecond (the speed of sound), hits an object and then bounces back. By measuring the time between the transmission of the pulse and the echo return, the distance to the object can be determined.



The SRF04 outputs a high-going pulse that corresponds to time required for the echo to return. PULSIN can be used to measure it and determine the distance to the target. There's a convenient side-effect with PULSIN on the BASIC Stamp 2: the value returned for the round-trip is in two microsecond units – the same as a one-way trip (sensor to target) in one microsecond units.


There are a couple of small technical details to be aware of, but otherwise, coding for the SRF04 is very straight forward. The trigger pulse must be at least 10 microseconds long. PULSOUT can do this for us. The other requirement is that we must wait 10 milliseconds between measurements.

The heart of this program is a subroutine called `Get_Sonar`. This routine started with the code sample that came with the sensor. While the Devantech example is perfectly suitable and easy to understand, the results seemed to bounce around a bit. Some software filtering (averaging several readings together) would smooth things out and make the output more useful.

The value returned by the routine will be stored in `rawDist`, so the code starts by clearing it. Then, within a loop, the code takes five readings from the sensor and averages them together. This may look a bit odd because most of us think about adding numbers and then dividing to get an average. We do the dividing first and then add the result into the return value because we could have an overflow if we do all the addition first. Yes, the dividing first technique can lead to rounding errors, but only if the values were very small. We didn't observe this when using the SRF04. Since one inch (the minimum range of the sensor) is about 74 microseconds, dividing by five (loop value) each time through causes no problem.

With the raw measurement complete, we can display it and convert it to units that make more sense to us humans. To convert the raw measurement to inches, we need to divide by 73.746. If we change this to 7.3746 then we'll have tenths of inches. Remember that we can't divide by a fractional number, but we can multiply by using the star-star (`**`) operator. Dividing by 7.3746 is the same as multiplying by 0.1356 (the reciprocal). To get the star-star parameter, we multiply 0.1356 by 65,536 to get 8886 (Note: Using 8886 with star-star is actually equally to multiplying by 0.13558 – pretty darned close). The same technique is used to convert the raw value to centimeters.

Since the units are in tenths, the value is divided by ten before sending to `RJ_Print`. On return the decimal place is printed, then the tenths digit by using the `DEC1` modifier.



```

=====
'
' File..... SONIC SIGHT.BS2
' Purpose... Devantech SRF04 Ultrasonic Range Finder
' Author.... Parallax
' E-mail.... stamptech@parallaxinc.com
' Started...
' Updated... 06 MAR 2002
'
' {$STAMP BS2}
'
=====

' -----
' Program Description
' -----

' This program uses the Devantech SRF04 to measure the distance between the
' unit and a target. Display is raw value, inches and centimeters.
'
' Conversion formulas:
'
' inches = echo_time / 73.746          (use 7.3746 for tenths)
' centimeters = echo_time / 29.033     (use 2.9033 for tenths)
'
' -----
' Revision History
' -----

' -----
' I/O Definitions
' -----

Trigger          CON      0
Echo              CON      1

' -----
' Constants
' -----

MoveTo            CON      2          ' cursor position control

' -----
' Variables
' -----

pWidth            VAR      Word        ' pulse width from sensor
rawDist           VAR      Word        ' filtered measurment
distance          VAR      Word        ' converted value
blips             VAR      Nib         ' loop counter for measurement

temp             VAR      Word        ' value for RJ_print
digits           VAR      Nib         ' used by RJ_Print

' -----
' EEPROM Data
' -----

' -----
' Initialization
' -----

Init:

```

```

PAUSE 250
DEBUG CLS
DEBUG "Devantech SRF04 Demo", CR
DEBUG "-----", CR, CR
DEBUG "Raw..... ", CR
DEBUG "Inches..... ", 34, CR
DEBUG "Centimeters... cm", CR

' -----
' Program Code
' -----

Main:
  GOSUB Get_Sonar          ' take sonar reading
  DEBUG MoveTo, 15, 3
  temp = rawDist
  GOSUB RJ_Print           ' display raw value

  DEBUG MoveTo, 15, 4
  distance = rawDist ** 8886 ' divide by 7.3746
  temp = distance / 10
  GOSUB RJ_Print           ' display inches
  DEBUG ".", DEC1 distance

  DEBUG MoveTo, 15, 5
  distance = rawDist ** 22572 ' divide by 2.9033
  temp = distance / 10
  GOSUB RJ_Print           ' display centimeters
  DEBUG ".", DEC1 distance

  PAUSE 200                ' delay between readings
  GOTO Main

END

' -----
' Subroutines
' -----

Get_Sonar:
  rawDist = 0
  FOR blips = 1 TO 5
    PULSOUT Trigger, 5      ' 10 uS trigger pulse
    PULSIN Echo, 1, pWidth  ' measure distance to target
    rawDist = rawDist + (pWidth / 5) ' simple digital filter
    PAUSE 10                ' minimum period between pulses
  NEXT
  RETURN

RJ_Print:                    ' right justify
  digits = 5
  LOOKDOWN temp, <[0,10,100,1000,65535], digits
  DEBUG REP " " \ (5 - digits), DEC temp
  RETURN

```